INTRODUCTION

Microchip’s PICmicro® microcontroller devices (PICs) have become a popular design choice for low-power and low-cost system solutions. The microcontrollers have multiple general-purpose input/output (GPIO) pins, and can be easily configured to implement Dallas Semiconductor’s 1-Wire® protocol. The 1-Wire protocol allows interaction with many Dallas Semiconductor parts including battery and thermal management, memory, iButtons®, and more. This application note will present general 1-Wire routines for a PIC16F628 and explain the timing and associated details. For added simplicity, a 4MHz clock is assumed for all material presented, and this frequency is available as an internal clock on many PICs. Appendix A of this document contains an include file with all 1-Wire routines. Appendix B presents a sample assembly code program designed for a PIC16F628 to read from a DS2761 High-Precision Li+ Battery Monitor. This application note is limited in scope to regular speed 1-Wire communication.

General Macros

In order to transmit the 1-Wire protocol as a master, only two GPIO states are necessary: high impedance and logic low. The following PIC assembly code snippets achieve these two states. The PIC16F628 has two GPIO ports, PORTA and PORTB. Either of the ports could be setup for 1-Wire communication, but for this example, PORTB is used. Also, the following code assumes that a constant DQ has been configured in the assembly code to indicate which bit in PORTB will be the 1-Wire pin. Throughout the code, this bit number is simply called DQ. Externally, this pin must be tied to a power supply via a pullup resistor.

```assembly
; Force the DQ line into a high impedance state.
OW_HIZ:MACRO
    BSF STATUS,RP0 ; Select Bank 1 of data memory
    BSF TRISB, DQ ; Make DQ pin High Z
    BCF STATUS,RP0 ; Select Bank 0 of data memory
    ENDM

; Force the DQ line to a logic low.
OW_LO:MACRO
    BCF STATUS,RP0 ; Select Bank 0 of data memory
    BCF PORTB, DQ ; Clear the DQ bit
    BSF STATUS,RP0 ; Select Bank 1 of data memory
    BCF TRISB, DQ ; Make DQ pin an output
    BCF STATUS,RP0 ; Select Bank 0 of data memory
    ENDM
```

Both of these snippets of code are written as macros. By writing the code as a macro, it is automatically inserted into the assembly source code by using a single macro call. This limits the number of times the code must be rewritten. The first macro, OW_HIZ, forces the DQ line to a high impedance state. The first step is to choose the bank 1 of data memory because the TRISB register is located in bank 1. Next, the DQ output driver is changed to a high impedance state by setting the DQ bit in the TRISB register. The last line of code changes back to bank 0 of data memory. The last line is not necessary, but is used so that all macros and function calls leave the data memory in a known state.

The second macro, OW_LO, forces the DQ line to a logic low. First, bank 0 of data memory is selected, so the PORTB register can be accessed. The PORTB register is the data register, and contains the values that will be forced to the TRISB pins if they are configured as outputs.

PICmicro is a registered trademark of Microchip Technology Inc.
1-Wire and iButton are registered trademarks of Dallas Semiconductor.

www.maxim-ic.com
The DQ bit of PORTB is cleared so the line will be forced low. Finally, bank 1 of data memory is selected, and the DQ bit of the TRISB register is cleared, making it an output driver. As always, the macro ends by selecting bank 0 of data memory.

A final macro labeled WAIT is included to produce delays for the 1-Wire signaling. WAIT is used to produce delays in multiples of 5μs. The macro is called with a value of TIME in microseconds, and the corresponding delay time is generated. The macro simply calculates the number of times that a 5μs delay is needed, and then loops within WAIT5U. The routine WAIT5U is shown in the next section. For each instruction within WAIT, the processing time is given as a comment to help understand how the delay is achieved.

```
WAIT:MACRO TIME
    ;Delay for TIME μs.
    ;Variable time must be in multiples of 5μs.
    MOVLW (TIME/5) - 1 ;1μs to process
    MOVWF TMP0  ;1μs to process
    CALL  WAIT5U ;2μs to process
ENDM
```

**General 1-Wire Routines**

The 1-Wire timing protocol has specific timing constraints that must be followed in order to achieve successful communication. To aid in making specific timing delays, the routine WAIT5U is used to generate 5μs delays. This routine is shown below.

```
WAIT5U:
    ;This takes 5μs to complete
    NOP  ;1μs to process
    NOP  ;1μs to process
    DECFSZ TMP0,F  ;1μs if not zero or 2μs if zero
    GOTO  WAIT5U ;2μs to process
    RETLW   0   ;2μs to process
```

When used in combination with the WAIT macro, simple timing delays can be generated. For example, if a 40μs delay is needed, WAIT 0.40 would be called. This causes the first 3 lines in WAIT to execute resulting in 4μs. Next, the first 4 lines of code in WAIT5U executes in 5μs and loops 6 times for a total of 30μs. The last loop of WAIT5U takes 6μs and then returns back to the WAIT macro. Thus, the total time to process would be 30 + 4 + 6 = 40μs.

**Table 1. Regular Speed 1-Wire Interface Timing**

| 2.5V ≤ VDD ≤ 5.5V, TA = -20°C to 70°C. |
|-------------------------------|----------------|---------|---------|---------|-------|
| **PARAMETER**      | **SYMBOL** | **MIN** | **TYP**  | **MAX**  | **UNITS** |
| Time Slot         | tSLOT     | 60      | 120      | µs       |
| Recovery Time     | tREC      | 1       |          | µs       |
| Write 0 Low Time  | tLOW0     | 60      | 120      | µs       |
| Write 1 Low Time  | tLOW1     | 1       | 15       | µs       |
| Read Data Valid   | tRDV      |         | 15       | µs       |
| Reset Time High   | tRSTH     | 480     |          | µs       |
| Reset Time Low    | tRSTL     | 480     | 960      | µs       |
| Presence Detect High | tPDH   | 15      | 60       | µs       |
| Presence Detect Low | tPDL | 60      | 240      | µs       |

The start of any 1-Wire transaction begins with a reset pulse from the master device followed by a presence detect pulse from the slave device. Figure 1 illustrates this transaction. This initialization sequence can easily be transmitted via the PIC, and the assembly code is shown below Figure 1. The 1-Wire timing specifications for initialization, reading, and writing are given above in Table 1. These parameters are referenced throughout the rest of the document.
**Figure 1. 1-Wire Initialization Sequence**

![Diagram of 1-Wire Initialization Sequence]

<table>
<thead>
<tr>
<th>LINE TYPE LEGEND:</th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Bus master active low</td>
<td>DS27XX active low</td>
</tr>
<tr>
<td>Both bus master and DS27XX active low</td>
<td>Resistor pullup</td>
</tr>
</tbody>
</table>

**OW_RESET:**

1. **OW_HIZ** ; Start with the line high
2. **CLRF PDBYTE** ; Clear the PD byte
3. **OW_LO**
4. **WAIT .500** ; Drive Low for 500μs
5. **OW_HIZ**
6. **WAIT .70** ; Release line and wait 70μs for PD Pulse
7. **BTFSS PORTB,DQ** ; Read for a PD Pulse
8. **INCF PDBYTE,F** ; Set PDBYTE to 1 if get a PD Pulse
9. **WAIT .430** ; Wait 430μs after PD Pulse
10. **RETLW 0**

The OW_RESET routine starts by ensuring the DQ pin is in a high impedance state so it can be pulled high by the pullup resistor. Next, it clears the PDBYTE register so it is ready to validate the next presence detect pulse. After that, the DQ pin is driven low for 500μs. This meets the tRSTL parameter shown in Table 1, and also provides a 20μs additional buffer. After driving the pin low, the pin is released to a high impedance state and a delay of 70μs is added before reading for the presence detect pulse. Using 70μs ensures that the PIC will sample at a valid time for any combination of tPDL and tPDH. Once the presence detect pulse is read, the PDBYTE register is adjusted to show the logic level read. The DQ pin is then left in a high-impedance state for an additional 430μs to ensure that the tRSTH time has been met, and includes a 20μs additional buffer.

The next routine needed for 1-Wire communication is DSTXBYTE, which is used to transmit data to a 1-Wire slave device. The PIC code for this routine is shown below Figure 2. This routine is called with the data to be sent in the WREG register, and it is immediately moved to the IOBYTE register. Next, a COUNT register is initialized to 8 to count the number of bits sent out the DQ line. Starting at the DSTXLP, the PIC starts sending out data. First the DQ pin is driven low for 5μs regardless of what logic level is sent. This ensures the tLOW1 time is met. Next, the lsb of the IOBYTE is shifted into the CARRY bit, and then tested for a one or a zero. If the CARRY is a one, the DQ bit of PORTB is set which changes the pin to a high impedance state and the line is pulled high by the pullup resistor. If the CARRY is a zero, the line is kept low. Next a delay of 60μs is added to allow for the minimum tLOW0 time. After the 60μs wait, the pin is changed to a high impedance state, and then an additional 2μs are added for pullup resistor recovery. Finally, the COUNT register is decremented. If the COUNT register is zero, all eight bits have been sent and the routine is done. If the COUNT register is not zero, another bit is sent starting at DSTXLP. A visual interpretation of the write zero and write one procedure is shown in Figure 2.
Figure 2. 1-Wire Write Time Slots

DSTXBYTE:    ; Byte to send starts in W
    MOVWF IOBYTE ; We send it from IOBYTE
    MOVLW .8
    MOVWF COUNT ; Set COUNT equal to 8 to count the bits

DSXRXL:     
    OW_LO
    NOP
    NOP
    NOP
    NOP
    NOP
    NOP
    RRF IOBYTE,F ; Drive the line low for 5μs
    BTFSC STATUS,C ; The data byte
    BSF PORTB,DQ ; Check the LSB of IOBYTE for 1 or 0
    WAIT .60 ; Drive the line high if LSB is 1
    OW_HIZ ; Continue driving line for 60μs
    NOP ; Release the line for pullup
    NOP
    NOP
    DECSZ COUNT,F ; Recovery time of 2μs
    GOTO DSTXRXL
    RETLW 0

The final routine for 1-Wire communication is DSXRBYTE, which allows the PIC to receive information from a slave device. The code is shown below Figure 3. The COUNT register is initialized to 8 before any DQ activity begins and its function is to count the number of bits received. The DSXRXL begins by driving the DQ pin low to signal to the slave device that the PIC is ready to receive data. The line is driven low for 6μs, and then released by putting the DQ pin into a high impedance state. Next, the PIC waits an additional 4μs before sampling the data line. There is 1 line of code in OW_LO after the line is driven low, and 3 lines of code within OW_HIZ. Each line takes 1μs to process. Adding up all the time results in 1 + 6 + 3 + 4 = 14μs which is just below the tRDV spec of 15μs. After the PORTB register is read, the DQ bit is masked off, and then the register is added to 255 to force the CARRY bit to mirror the DQ bit. The CARRY bit is then shifted into IOBYTE where the incoming byte is stored. Once the byte is stored a delay of 50μs is added to ensure that tSLOT is met. The last check is to determine if the COUNT register is zero. If it is zero, 8 bits have been read, and the routine is exited. Otherwise, the loop is repeated at DSXRXL. The read zero and read one transactions are visually shown in Figure 3.
DSRXBYTE:    ; Byte read is stored in IOBYTE  
  MOVLW .8  
  MOVWF COUNT ; Set COUNT equal to 8 to count the bits  
DSRXLp:      
  OW_LO      ; Bring DQ low for 6μs  
  NOP        
  NOP        
  NOP        
  NOP        
  NOP        
  NOP        
  NOP        
  OW_HIZ     ; Change to HiZ and Wait 4μs  
  NOP        
  NOP        
  NOP        
  NOP        
  MOVF PORTB,W ; Read DQ  
  ANDLW 1<<DQ ; Mask off the DQ bit  
  ADDLW .255 ; C = 1 if DQ = 1; C = 0 if DQ = 0  
  RRF IOBYTE,F ; Shift C into IOBYTE  
  WAIT .50 ; Wait 50μs to end of time slot  
  DECFSZ COUNT,F ; Decrement the bit counter  
  GOTO DSRXLp 
  RETLW 0

SUMMARY

Dallas Semiconductor’s 1-Wire communication protocol can easily be implemented on Microchip’s PICmicro line of microcontrollers. In order to complete 1-Wire transactions, only two GPIO states are needed, and the multiple GPIOs on a PIC are easily configured for this task. There are three basic routines necessary for 1-Wire communication: Initialization, Read Byte, and Write Byte. These three routines have been presented and thoroughly detailed to provide accurate 1-Wire regular speed communication. This allows a PIC to interface with any of the many Dallas Semiconductor 1-Wire devices. Appendix A of this document has all three routines in a convenient include file. Appendix B contains a small assembly program meant to interface a PIC16F628 to a DS2761 High-Precision Li+ Battery Monitor.
APPENDIX A: 1-WIRE INCLUDE FILE (1W_16F6X.INC)

; *******************************************************
; Dallas 1-Wire Support for PIC16F628
; Processor has 4MHz clock and 1\mu s per instruction cycle.
; *******************************************************

; *******************************************************
; Dallas Semiconductor 1-Wire MACROS
; *******************************************************

OW_HIZ:MACRO
    BSF STATUS,RP0  ; Select Bank 1 of data memory
    BSF TRISB, DQ  ; Make DQ pin High Z
    BCF STATUS,RP0  ; Select Bank 0 of data memory
ENDM

; --------------------------------------------------------

OW_LO:MACRO
    BCF STATUS,RP0  ; Select Bank 0 of data memory
    BCF PORTB, DQ  ; Clear the DQ bit
    BSF STATUS,RP0  ; Select Bank 1 of data memory
    BCF TRISB, DQ  ; Make DQ pin an output
    BCF STATUS,RP0  ; Select Bank 0 of data memory
ENDM

; *******************************************************

WAIT:MACRO TIME
; Delay for TIME \mu s.
; Variable time must be in multiples of 5\mu s.
    MOVLW (TIME/5)-1  ;1\mu s
    MOVWF TMP0  ;1\mu s
    CALL WAIT5U  ;2\mu s
ENDM

; *******************************************************

; Dallas Semiconductor 1-Wire ROUTINES
; *******************************************************

WAIT5U:
; This takes 5\mu s to complete
    NOP  ;1\mu s
    NOP  ;1\mu s
    DECFSZ TMP0,F  ;1\mu s or 2\mu s
    GOTO WAIT5U  ;2\mu s
    RETLW 0   ;2\mu s

; *******************************************************

OW_RESET:
    OW_HIZ       ; Start with the line high
    CLRF PDBYTE  ; Clear the PD byte
    OW_LO
    WAIT .500  ; Drive Low for 500\mu s
    OW_HIZ
    WAIT .70   ; Release line and wait 70\mu s for PD Pulse
    BTFSS PORTB,DQ  ; Read for a PD Pulse
    INCF PDBYTE,F  ; Set PDBYTE to 1 if get a PD Pulse
WAIT .400 ; Wait 400μs after PD Pulse
RETLW 0

; --------------------------------------------------------

DSRXBYTE:    ; Byte read is stored in IOBYTE
   MOVWL .8
   MOVWF COUNT ; Set COUNT equal to 8 to count the bits

DSRXLP:
   OW_LO
   NOP
   NOP
   NOP
   NOP
   NOP
   NOP
   NOP
   ; Bring DQ low for 6μs
   OW_HIZ
   NOP
   NOP
   NOP
   NOP
   NOP
   ; Change to HiZ and Wait 4μs
   MOVF PORTB,W ; Read DQ
   ANDLW 1<<DQ ; Mask off the DQ bit
   ADDLW .255 ; C=1 if DQ=1:  C=0 if DQ=0
   RRF IOBYTE,F ; Shift C into IOBYTE
   WAIT .50 ; Wait 50μs to end of time slot
   DECFSZ COUNT,F ; Decrement the bit counter
   GOTO DSRXLP
   RETLW 0

; --------------------------------------------------------

DSTXBYTE:    ; Byte to send starts in W
   MOVWF IOBYTE ; We send it from IOBYTE
   MOVWL .8
   MOVWF COUNT ; Set COUNT equal to 8 to count the bits

DSTXLP:
   OW_LO
   NOP
   NOP
   NOP
   NOP
   NOP
   NOP
   ; Drive the line low for 5μs
   RRF IOBYTE,F ; The data byte
   BTFSC STATUS,C ; Check the LSB of IOBYTE for 1 or 0
   BSF PORTB,DQ ; Drive the line high if LSB is 1
   WAIT .60 ; Continue driving line for 60μs
   OW_HIZ
   NOP
   ; Recovery time of 2μs
   DECFSZ COUNT,F ; Decrement the bit counter
   GOTO DSTXLP
   RETLW 0

; --------------------------------------------------------
APPENDIX B: PIC16F628 TO DS2761 ASSEMBLY CODE (PIC_2_1W.ASM)

; *******************************************;
; Dallas Semiconductor PIC code
; This code will interface a PIC16F628 microcontroller to
a DS2761 High-Precision Li+ Battery Monitor
; *******************************************;

VCC
  ^
   |
   |
  \ Rpup
   / 
  /  
 /  
|  
|  
|  
16F628 | DS2761
|  
|  
|
|  
|
|  
|
; 16F628          | DS2761
; RB1 (pin 7) --------------- DQ (pin 7)
; *******************************************;

; List your processor here.
list p=16F628

; Include the processor header file here.
#include <p16F628.inc>

; Assign the PORTB with Constants
constant DQ=1          ; Use RB1 (pin7) for 1-Wire

; These constants are standard 1-Wire ROM commands
constant SRCHROM=0xF0
constant RDROM=0x33
constant MTCHROM=0x55
constant SKPROM=0xCC

; These constants are used throughout the code
cblock  0x20
IOBYTE
TMP0   ; Address 0x23
COUNT   ; Keep track of bits
PICMSB  ; Store the MSB
PICLSB  ; Store the LSB
PDBYTE  ; Presence Detect Pulse
endc

; Setup your configuration word by using __config.
; For the 16F628, the bits are:
; CP1,CP0,CP1,CP0,N/A, CPD, LVP, BODEN, MCLRE, FOSC2, PWRTE, WDTE, FOSC1, FOSC0
; CP1 and CP0 are the Code Protection bits
; CPD: is the Data Code Protection Bit
; LVP is the Low Voltage Programming Enable bit
; PWRTE is the power-up Timer enable bit
; WDTE is the Watchdog timer enable bit
; FOSC2, FOSC1 and FOSC0 are the oscillator selection bits.

; CP disabled, LVP disabled, BOD disabled, MCLR enabled, PWRT disabled, WDT disabled, INTRC I/O oscillator
; 1111100111000

;__config 0x3F38

; Set the program origin for subsequent code.

org 0x00
GOTO SETUP
NOP
NOP
GOTO INTERRUPT ; PC 0x04...INTERRUPT VECTOR!

INTERRUPT:
SLEEP

; Option Register bits
;
; RBPU,INTEDG,TOCS,TOSE,PSA,PS2,PS1,PS0
; 7=PORTB Pullup Enable, 6=Interrupt Edge Select, 5=TMR0 Source,
; 4=TMR0 Source Edge, 3=Prescaler Assign, 2-0=Prescaler Rate Select

; 11010111
; PORTB pullups disabled,rising edge,internal,hightolow,TMR0,1:256

SETUP:
BCF STATUS,RP1
BSF STATUS,RP0 ; Select Bank 1 of data memory
MOVLW 0xD7
MOVWF OPTION_REG
BCF STATUS,RP0 ; Select Bank 0 of data memory

BCF INTCON,7 ; Disable all interrupts.

GOTO START

; Include the 1-Wire communication routines and macros

#include 1w_16f6x.inc

START:

GET_TEMP:
CALL OW_RESET ; Send Reset Pulse and read for Presence Detect Pulse
BTFSS PDBYTE,0 ; 1 = Presence Detect Detected
GOTO NOPDPULSE
MOVLW SKPROM
CALL DSTXBYTE ; Send Skip ROM Command (0xCC)
MOVLW 0x69
CALL DSTXBYTE ; Send Read Data Command (0x69)
MOVLW 0x0E
CALL DSTXBYTE ; Send the DS2761 Current Register MSB address (0x0E)
CALL DSRXBYTE ; Read the DS2761 Current Register MSB
MOVF IOBYTE,W
MOVWF PICMSB ; Put the Current MSB into file PICMSB
CALL DSRXBYTE ; Read the DS2761 Current Register LSB
MOVF IOBYTE,W
MOVWF PICLSB ; Put the Current LSB into file PICLSB
CALL OW_RESET

NOPDPULSE: ; Add some error processing here!
SLEEP ; Put PIC to sleep

;---------------------------------------------------------
end